
Disparity selectivity of Simple Cells

This Mathematica notebook shows how a disparity tuning curve may be
obtained from left and right receptive field profiles of simple cells.
Curves obtained in this notebook have been presented in:

DeAngelis, G.C., Ohzawa, I., and Freeman, R.D. Neural mechanisms
underlying stereopsis:  how do simple cells in the visual cortex encode
binocular disparity?  Perception,  24: 3-31, 1995  (Fig. 2)

Ohzawa, I., DeAngelis, G.C., and Freeman, R.D.  The neural coding of
stereoscopic depth,  NeuroReport,  1997  (Fig. 1)

Pos[] is a half-wave rectifier function.  This is to model simple cell’s
firing property.

Pos[x_] := Max[0, x]

IBinoc[] is binocular receptive field interaction function for identical LR
receptive fields located at the corresponding points, 
where x is the space variable, and d is the binocular disparity.

IBinoc[x_, d_] :=  (Exp[-5.5 x^2] Cos[2 Pi x]
+ Exp[-5.5 (x-d)^2] Cos[2 Pi (x-d)])

Binoc response.  Axis left-bottom is x, while that going from bottom
to upper-right is d.  Threshold is also a factor here.

Plot3D[ Pos[IBinoc[x, d] -0.75],

{x, -1, 1}, {d, -1, 1},
PlotPoints -> 40,
PlotRange -> {0, 1.3}
]

-SurfaceGraphics-

QBinoc[] is binocular receptive field interaction function for quadrature LR
receptive fields located at the corresponding points, 
where x is the space variable, and d is the binocular disparity.

QBinoc[x_, d_] :=  (Exp[-5.5 x^2] Cos[2 Pi x]
- Exp[-5.5 (x-d)^2] Sin[2 Pi (x-d)])

Binoc response.  Axis left-bottom is x, while that going from bottom
to upper-right is d.  Threshold is also a factor here.

Plot3D[ Pos[QBinoc[x, d] -0.5],

{x, -1, 1}, {d, -1, 1},
PlotPoints -> 40,
PlotRange -> {0, 1.5}
]

Disparity tuning curve can be obtained by integrating the plot above for x,
for each d value.  This is the case for matched RF located at corresponding
points.

Plot[ NIntegrate[ Pos[IBinoc[x, d]-0.75],{x, -1, 1} ],

{d, -1, 1}
]

Disparity tuning curve can be obtained by integrating the plot above for x,
for each d value.  This is the case for quadrature RFs located at corresponding
points.

Plot[ NIntegrate[ Pos[QBinoc[x, d]-0.5],{x, -1, 1} ],

{d, -1, 1}
]

Binoc response for quadrature LR receptive field combination again,
but try it with the same threshold as in-phase version.  Note that the
right eye response (sine-phase RF) disappears into the threshold.

Plot3D[ Pos[QBinoc[x, d] -0.75],

{x, -1, 1}, {d, -1, 1},
PlotPoints -> 40,
PlotRange -> {0, 1.3}
]

Disparity tuning curve for quadrature LR fields with somewhat deeper
firing threshold.

Plot[ NIntegrate[ Pos[QBinoc[x, d]-0.75],{x, -1, 1} ],

{d, -1, 1}
]

InvBinoc[x_, d_] :=  (Exp[-5.5 x^2] Cos[2 Pi x]
- Exp[-5.5 (x-d)^2] Cos[2 Pi (x-d)])

Plot3D[ Pos[InvBinoc[x, d] -0.75],

{x, -1, 1}, {d, -1, 1},
PlotPoints -> 40,
PlotRange -> {0, 1.0}
]

